

CÁC PHÉP TOÁN MỘT SỐ HÀM TÍNH TOÁN, RÚT GỌN KHÁC.

I. Các phép toán cơ bản (cộng, trừ, nhân, chia, lũy thừa, khai căn,...)

1) Phép cộng, trừ, nhân, chia, khai căn,...

Năm phép toán ‘cộng’, ‘trừ’, ‘nhân’, ‘chia’, ‘lũy thừa’ được khai báo với các kí hiệu tương ứng là ‘+’, ‘-’, ‘*’, ‘/’, ‘^’.

Riêng phép khai căn phải dùng cú pháp khác.

+ căn bậc hai của số a (thực hay phức) được khai báo bằng lệnh:

> sqrt(a); hoặc > root(a, 2);

+ căn bậc n của số a (thực hay phức) được khai báo bằng lệnh:

> surd(a, n); hoặc > root(a, n);

Các ví dụ:

> 2+1/2;

$$\frac{5}{2}$$

> x-3/5;

$$x - \frac{3}{5}$$

> 12*5/7-3/2;

$$\frac{99}{14}$$

> 2^6;

$$64$$

> (1/3+5)^4-753;

$$\frac{4543}{81}$$

Chú ý: Maple có thể tính toán được các số rất lớn. Chẳng hạn:

> 12345^(234);

25639884158519854915018134511107491129671081010129089960626926808606062036226435543712552\19426949823525337150206736282506477711337552743322863049645616134432668416022992862178314\80466581128516178692725431042462153498499367419918717426831428560000341117658718243119146\62587798488004853209015894768913279077394429411238061049146198115936339008763402759771394\71910783173844656842091584544650330257074132153976061975134830297704833002309648009883165\26637109986579755844611433159475092495697177193773851181429263639274242701126530281939259\09842565205899639659430803166780485636182512452690475249996610105878815264134692460138955\12231379578446094318735037357379270492616140861041921838012603752480347193072377297540669\92408518624848093164989752250601796566817769684737857434248482333154320777465277172914052\04585787987812464861185347714698688218569918180240208034252374305779022091608079482871729\68578514172950935396866540584570604011105388053692877292633056640625

Số chữ số của số trên có thể tìm được bằng lệnh:

> length(%);

958

Maple có thể tính toán trên các biểu thức chứa biến:

> **3*x+5*y-2+x-y;**

$$4x + 4y - 2$$

> **(x+y)/(x^2-y^2);**

$$\frac{x+y}{x^2-y^2}$$

Để rút gọn biểu thức trên ta dùng lệnh như sau:

> **simplify(%);**

$$-\frac{1}{y-x}$$

* Tính $\sqrt{12345}$:

> **sqrt(12345);**

$$\sqrt{12345}$$

+ Tính $\sqrt{123468}$:

> **sqrt(123468);**

$$2\sqrt{30867}$$

+ Tính $\sqrt[3]{-8}$:

- Nếu dùng lệnh **surd** thì được kết quả:

> **surd(-8,3);**

$$-2$$

- Nếu dùng lệnh **root** thì được kết quả:

> **root(-8,3);**

$$2(-1)^{\left(\frac{1}{3}\right)}$$

+ Tính $\sqrt[3]{-8,0}$:

- Nếu dùng lệnh **surd** thì được kết quả:

> **surd(-8.0,3);**

$$-2.000000000$$

- Nếu dùng lệnh **root** thì được kết quả:

> **root(-8.0,3);**

$$1.000000000 + 1.732050807I$$

2) Gán giá trị cho một biểu thức hay một biến.

Gán giá trị của f(biểu thức, hàm, thủ tục, list,...) cho **a** ta dùng cú pháp:

Cú pháp: > **a: = f;**

Ví dụ 1:

Gán giá trị của kết quả $\sqrt{24} + \frac{1}{4}$ vào biến a , ta thực hiện như sau:

> **a:=sqrt(24)+1/4;**

$$a := 2\sqrt{6} + \frac{1}{4}$$

Sau khi gán giá trị vào một biến, ta có thể dùng các hàm, thủ tục khác để tác động lên biến đó.

Chẳng hạn với a được xác định như ví dụ trên ta có thể tính a^2 .

> **`a^2`:=a^2;**

$$a^2 := \left(2\sqrt{6} + \frac{1}{4}\right)^2$$

II. Một số hàm trên các phép toán.

1) Kết hợp phép tính căn $\sqrt{a}.\sqrt{b}$ thành \sqrt{ab} .

Cú pháp: [> **combine(...)** ;

Thường thì khi nhập $\sqrt{a}.\sqrt{b}$ thì Maple vẫn hiển thị kết quả như khi nhập.

Xem các ví dụ:

Nhập vào Maple $\sqrt{3}.\sqrt{6}$:

> **sqrt(3)*sqrt(6);**

$$\sqrt{3} \sqrt{6}$$

+ Để nhân hai căn thức này ta dùng lệnh:

> **combine(%);**

$$\sqrt{18}$$

+ Để đổi căn số trên ra số thập phân ta dùng lệnh sau:

> **evalf(%);**

$$4.242640687$$

2. Trục căn thức ở mẫu của biểu thức chứa căn.

Cú pháp: > **rationalize(a)** ;

Ví dụ 1:

Cho biểu thức $a = \frac{1}{2 + \sqrt{3} + \sqrt{2}}$.

+Nhập a vào Maple:

> **a:=1/(2+sqrt(3)+sqrt(2));**

$$a := \frac{1}{2 + \sqrt{3} + \sqrt{2}}$$

+ Trục căn thức ở mẫu của a :

> **rationalize(a);**

$$\frac{(2 + \sqrt{2} - \sqrt{3})(-3 + 4\sqrt{2})}{23}$$

Ví dụ 2:

Trục căn thức ở mẫu của biểu thức $p = \frac{1}{\sqrt{x+2}}$

+ Nhập vào Maple:

> **p:=1/(sqrt(x)+2);**

$$p := \frac{1}{\sqrt{x} + 2}$$

+ Trục căn thức ở mẫu:

> **p:=rationalize(p);**

$$p := \frac{-2 + \sqrt{x}}{-4 + x}$$

3. Đơn giản một biểu thức chứa căn.

Cú pháp: > **simplify(a, 'option');**

Trong đó: - a là biểu thức chứa căn;

- option là các từ khóa như 'radical', 'sqrt', 'power',...

Ví dụ 1:

Cho biểu thức $a = \frac{1}{\sqrt{2}} + 3 - \frac{2}{\sqrt{2}+1}$.

+ Nhập biểu thức trên vào Maple:

> **a:=1/sqrt(2)+3-2/(sqrt(2)+1);**

$$a := \frac{\sqrt{2}}{2} + 3 - \frac{2}{\sqrt{2} + 1}$$

+ Đơn giản biểu thức trên:

> **simplify(a);**

$$\frac{4 + 7\sqrt{2}}{2(\sqrt{2} + 1)}$$

+ Trục căn thức ở mẫu, ta được

> **a:=rationalize(a);**

$$a := \frac{(4 + 7\sqrt{2})(-1 + \sqrt{2})}{2}$$

+ Khai triển (bằng lệnh **expand**) ta được:

> **a:=expand(a);**

$$a := 5 - \frac{3\sqrt{2}}{2}$$

Ví dụ 2:

Cho biểu thức $b = \sqrt{2}\sqrt{6} - 2\sqrt{2}\sqrt{x+1}$.

+Nhập biểu thức vào Maple:

> **b:=sqrt(2)*sqrt(6)-2*sqrt(2)*sqrt(x+1);**

$$b := \sqrt{2} \sqrt{6} - 2 \sqrt{2} \sqrt{x+1}$$

+ Nếu dùng lệnh **simplify** ta được kết quả:

> **b:=simplify(b);**

$$b := 2 \sqrt{3} - 2 \sqrt{2} \sqrt{x+1}$$

+ Nếu muốn gộp các dấu căn ta phải dùng lệnh **combine**:

> **b:=combine(b,radical);**

$$b := 2 \sqrt{3} - 2 \sqrt{2x+2}$$

Ví dụ 3:

Cho biểu thức $c = \sqrt{x^3} \cdot \sqrt{x+1} - 2\sqrt{6}\sqrt{y}\sqrt{2x}$.

+ Nhập biểu thức vào Maple:

> **a:=sqrt(x)^3*sqrt(x+1)-2*sqrt(6)*sqrt(y)*sqrt(2*x);**

$$a := x^{\left(\frac{3}{2}\right)} \sqrt{x+1} - 2 \sqrt{6} \sqrt{y} \sqrt{2} \sqrt{x}$$

+ Đơn giản c bằng lệnh simplify ta được kết quả:

> **a:=simplify(a);**

$$a := x^{\left(\frac{3}{2}\right)} \sqrt{x+1} - 4 \sqrt{3} \sqrt{y} \sqrt{x}$$

+ Làm gọn c bằng lệnh combine (c, radical) ta được kết quả:

> **c:=combine(c,radical);**

$$a := x^{\left(\frac{3}{2}\right)} \sqrt{x+1} - 4 \sqrt{x} \sqrt{3} \sqrt{y}$$

+ Nếu khai báo thêm từ khóa “symbolic” trong câu lệnh, ta được kết quả:

> **a:=combine(a,radical,symbolic);**

$$a := x \sqrt{x(x+1)} - 4 \sqrt{3} \sqrt{xy}$$

4. Biến đổi các biểu thức chứa căn về dạng đơn giản.

Cú pháp: > **radnormal(a);**

Ví dụ 1:

Cho biểu thức $a = (4 - 3\sqrt{2})^3$.

+Nhập a vào Maple:

> **a:=(4-3*sqrt(2))^3;**

$$a := (4 - 3\sqrt{2})^3$$

+ Nếu đơn giản a bằng lệnh simplify, ta được kết quả:

> **a:=simplify(a);**

$$a := -(-4 + 3\sqrt{2})^3$$

+ Cũng dùng simplify nhưng thêm option 'radical', ta được kết quả:

> **a:=simplify(a, radical);**

$$a := (4 - 3\sqrt{2})^3$$

+ Nếu khai triển dùng lệnh 'expand', ta được kết quả:

> **a:=expand(a);**

$$a := 280 - 198\sqrt{2}$$

+ Nếu dùng lệnh 'radnormal' ta cũng được kết quả:

> **a:=radnormal(a);**

$$a := 280 - 198\sqrt{2}$$

Ví dụ 2:

Cho biểu thức $a = \sqrt{59 + 30\sqrt{2}}$.

+Nhập vào Maple:

> **a:=(59+30*sqrt(2))^(1/2);**

$$a := \sqrt{59 + 30\sqrt{2}}$$

+ Nếu khai triển bằng lệnh 'expand', ta được kết quả:

> **a:=expand(a);**

$$a := \sqrt{59 + 30\sqrt{2}}$$

+ Nếu dùng lệnh 'radnormal', ta được kết quả:

> **a:=radnormal(a);**

$$a := 3 + 5\sqrt{2}$$

Ví dụ 4:

Cho biểu thức $b = \sqrt{20x^2 + 120x + 180}$.

+Nhập biểu thức vào Maple:

> **b:=(20*x^2+120*x+180)^(1/2);**

$$b := \sqrt{20x^2 + 120x + 180}$$

+ Đơn giản bằng lệnh 'radnormal', ta được:

> **b:=radnormal(b);**

$$b := 2\sqrt{5}\sqrt{(x+3)^2}$$

+ Đơn giản bằng lệnh 'simplify' có option 'sqrt', ta được:

> **b:=simplify(b,sqrt);**

$$b := 2\sqrt{5}\operatorname{csgn}(x+3)(x+3)$$

+ Nếu giả sử $x > -3$ trước khi dùng lệnh trên, ta được:

```
> assume(x>-3);
b:=simplify(b,sqrt);
```

$$b := 2\sqrt{5}(x+3)$$

+ Nếu đơn giản biểu thức với 'RealDomain', ta được:

```
> use RealDomain in simplify(b) end use;
```

$$2\sqrt{5}|x+3|$$

Chú ý: Với lệnh trên mà ta vẫn khai báo option 'sqrt' trong simplify thì kết quả là:

```
> use RealDomain in simplify(b,sqrt) end use;
```

$$2\sqrt{5} \operatorname{signum}(x+3)(x+3)$$

♣ Mong rằng, qua một số ví dụ trên quý bạn đọc nhận biết được chức năng cụ thể của mỗi lệnh để vận dụng có hiệu quả. Đặc biệt là tính 'gắn gói' của các lệnh rất khó phân biệt và các 'option' trong mỗi lệnh.

5) Hàm tính giá trị tuyệt đối của số thực x.

Cú pháp: **> abs(x);**

Ví dụ 1:

```
> abs(2*x);
```

$$2|x|$$

```
> abs(sqrt(3+2*sqrt(2))*x^2);
```

$$(\sqrt{2}+1)|x|^2$$

```
> a:=abs(-5*x*y*z^2);
```

$$a := 5|xyz^2|$$

Có thể dùng hàm 'expand' để phân tích kết quả trên:

```
> a:=expand(a);
```

$$a := 5|x||y||z|^2$$

Ví dụ 2:

Cho biểu thức $m = 5|x| \cdot |x-1| \cdot |y|^2$

+ Nhập vào Maple:

```
> m:=5*abs(x)*abs(x-1)*abs(y^2);
```

$$m := 5|x||x-1||y|^2$$

+ Có thể kết hợp các giá trị tuyệt đối trong kết quả trên dùng hàm 'combine' cùng với option 'abs':

```
> m:=combine(m,abs);
```

$$m := 5|x(x-1)y^2|$$

6. Hàm cộng một chuỗi các giá trị.

a) Cú pháp 1: **> add(i, i=m..n);**

b) Cú pháp 2: `> add(i, i=[1..6]);`

c) Cú pháp 3: `> add(i, i in x);`

Trong đó: - m, n là các số nguyên dương;
- [li] là một tập hợp các số (thực);
- x là một biểu thức.

Ví dụ 1:

Tính tổng các số i^3 biết $i \in \{1; 2; \dots; 6\}$.

+ Nhập biểu thức tính vào Maple:

```
> add(i^3, i=1..6);  
441
```

Có thể xây dựng tập hợp $\{1; 2; \dots; 6\}$ sau đó tính tổng của i^3 trong tập đó:

+ Xây dựng tập hợp:

```
> x:= [seq(i, i=1..6)];  
x := [1, 2, 3, 4, 5, 6]
```

+ Tính tổng các số i^3 với i thuộc tập hợp đó:

```
> add(i^3, i = x);  
441
```

+ Cũng có thể dùng câu lệnh thứ ba như sau:

```
> add(i^3, i in x);  
441
```

Vận dụng hàm này để xây dựng chương trình tính chu vi của một tam giác.

Ví dụ 2:

Trong mặt phẳng với hệ trục tọa độ Oxy cho ba điểm $A(1; 2), B(-1; 1), C(0; -2)$.

Tính chu vi của tam giác ABC.

+ Nhập các điểm A, B, C vào Maple và xác định tam giác ABC:

```
> with(geometry):  
> point(A,1,1), point(B,-1,1), point(C,0,-2):  
> triangle(ABC,[A,B,C]);  
ABC
```

+ Tính độ dài các cạnh của tam giác ABC:

```
> canh:=sides(ABC);  
canh := [ $\sqrt{4}$ ,  $\sqrt{10}$ ,  $\sqrt{10}$ ]
```

Làm gọn kết quả:

```
> canh:=simplify(canh);  
canh := [2,  $\sqrt{10}$ ,  $\sqrt{10}$ ]
```

+ Tính chu vi của tam giác ABC:

```
> add(i, i in canh);  
2 + 2 $\sqrt{10}$ 
```


Ví dụ 5:

Viết đa thức có dạng $P(x) = a_0 + a_1x + a_2x^2 + \dots + a_8x^8$.

+ Dùng Maple để viết đa thức trên:

> **P(x):=add(a[i]*x^i, i=0..8);**

$$P(x) := a_0 + a_1x + a_2x^2 + a_3x^3 + a_4x^4 + a_5x^5 + a_6x^6 + a_7x^7 + a_8x^8$$

Chú ý: Hàm add chỉ có thể tính tổng trong khoảng chứa số cụ thể. Một khoảng chưa xác định (chứa biến) thì Maple không thực hiện được lệnh add.

Chẳng hạn, tính tổng $1 + 2 + 3 + \dots + n$:

> **add(i, i=1..n);**

Error, unable to execute add

Ta thấy, Maple đã thông báo lỗi và không thể thực hiện lệnh 'add'.

♣ Để khắc phục nhược điểm này ta có thể dùng hàm 'sum' như sau:

> **S:=Sum(i, i=1..n);**

$$S := \sum_{i=1}^n i$$

> **S:=value(S);**

$$S := \frac{(n+1)^2}{2} - \frac{n}{2} - \frac{1}{2}$$

(Hàm sum sẽ được giới thiệu sau)

7. Hàm nhân một chuỗi các giá trị.

a) Cú pháp 1: > **mul(i, i=m..n);**

b) Cú pháp 2: > **mul(i, i=[li]);**

c) Cú pháp 3: > **mul(i, i in x);**

Trong đó: - m, n là các số nguyên dương;
- [li] là một tập hợp các số (thực);
- x là một biểu thức.

Ví dụ 1:

Tính tích của 7 số tự nhiên đầu tiên (khác không):

+ Nhập biểu thức tính vào Maple:

> **mul(i, i=1..7);**

5040

Ví dụ 2:

Tính tích $P = \left(1 + \frac{1}{2}\right) \left(1 + \frac{1}{3}\right) \dots \left(1 + \frac{1}{2008}\right)$.

+Nhập biểu thức tính vào Maple:

> **mul(1+1/i, i=2..2008);**

$\frac{2009}{2}$

8. Hàm sum , hàm product.

Các ví dụ:

Ví dụ 1:

Tính tổng $S_1 = 1 + 3 + 5 + \dots + (2n + 1)$.

+ Nhập biểu thức vào Maple:

> **S1:=Sum(2*i+1, i=0..n);**

$$S1 := \sum_{i=0}^n (2i + 1)$$

+ Tính giá trị của biểu thức trên:

> **S1:=value(S1);**

$$S1 := (n + 1)^2$$

Nhận xét: Trong câu lệnh nhập biểu thức ta dùng hàm ‘Sum’ _ kí tự S viết hoa.

Maple quy định, hàm ‘Sum’ cho hiện biểu thức của tổng cần xác định. Còn hàm ‘sum’ _ chữ s viết thường _ tính giá trị của tổng cần xác định.

Chẳng hạn với tổng ở Ví dụ 1 trên, ta dùng hàm ‘sum’ thì được kết quả:

> **S1:=sum(2*i+1, i=0..n);**

$$S1 := (n + 1)^2$$

Ví dụ 2:

Tìm tổng các nghiệm của phương trình $2x^3 - 3x^2 + x - 1 = 0$.

+Để giải bài toán trên ta tìm tổng của các số k với k thuộc tập nghiệm của phương trình trên (dùng hàm RootOf() để xác định tập nghiệm của phương trình):

> **sum(k, k=RootOf(2*x^3-3*x^2+x-1));**

$$\frac{3}{2}$$

+ Tìm tổng bình phương của các nghiệm của phương trình trên:

> **sum(k^2, k=RootOf(2*x^3-3*x^2+x-1));**

$$\frac{5}{4}$$

+ Tìm tổng lập phương các nghiệm của phương trình trên:

> **sum(k^3, k=RootOf(2*x^3-3*x^2+x-1));**

$$\frac{21}{8}$$

{Nếu dùng hàm ‘Sum’ thì kết quả là:

> **Sum(k^3, k=RootOf(2*x^3-3*x^2+x-1));**

$$\sum_{k = \text{RootOf}(2_Z^3 - 3_Z^2 + _Z - 1)} k^3$$

+Xem kết quả bằng lệnh:

> **value(%);**

$$\frac{21}{8}$$

+Nếu dùng hàm ‘evalf’ để xem kết quả thì ta được:

> **evalf(%)**;

$$2.625000000 + 0.1$$

Ta nhận thấy kết quả của hàm ‘evalf’ không trực quan bằng kết quả của ‘value’.

Qua đây chúng tôi mong quý bạn đọc lưu ý cách dùng hàm ‘value’ hoặc ‘eval’ một cách phù hợp để tính giá trị của một kết quả nào đó}.

Ví dụ 3:

Tính tích các nghiệm số của phương trình $7x^5 - 4x^2 + 3x - 5 = 0$

+ Lệnh tính tích các nghiệm:

> **product(x, x=RootOf(7*x^5-4*x^2+3*x-5))**;

$$\frac{5}{7}$$

Ví dụ 4:

Xây dựng biểu thức $f(x) = (x+1)(x+2)(x+3)(x+4)$.

+ Nhập vào Maple:

> **f(x):=product(x+i, i=1..4)**;

$$f(x) := (x+1)(x+2)(x+3)(x+4)$$

⊗ Nếu dùng thủ tục ‘Product’ thì kết quả hiển thị dưới dạng *kí hiệu*.

Chẳng hạn:

> **Product(i+1, i=1..n)**;

$$\prod_{i=1}^n (i+1)$$

Ví dụ 5:

Gọi x_1, x_2 là hai nghiệm của phương trình $x^2 - \sqrt{2}x - \sqrt{3} + 1 = 0$

Tính giá trị của biểu thức $Q = \frac{1}{x_1^2 + x_2^2} - \frac{x_1 + x_2}{3x_1x_2}$.

+ Xác định tập nghiệm của phương trình dùng lệnh RootOf:

> **T:=RootOf(x^2-sqrt(2)*x-sqrt(3)+1)**;

$$T := \text{RootOf}(_Z^2 - \sqrt{2} _Z - \sqrt{3} + 1)$$

+ Tính giá trị của tổng $x_1^2 + x_2^2$:

> **s1:=sum(x^2, x=T)**;

$$s1 := 2 \text{RootOf}(_Z^2 - 3, \text{index} = 1)$$

Kết quả hiển thị chưa cụ thể. Để xem chi tiết giá trị của s1, ta dùng lệnh:

> **s1:=allvalues(%)**;

$$s1 := 2\sqrt{3}$$

{Nếu dùng lệnh ‘value’ hay ‘evalf’ thì kết quả không cụ thể bằng lệnh ‘allvalue’}

+ Tính giá trị của tổng $x_1 + x_2$:

> **s2:=sum(x, x=T)**;

s2:=allvalues(%);

$$s2 := \text{RootOf}(_Z^2 - 2, \text{index} = 1)$$

$$s2 := \sqrt{2}$$

+ Tính tích $x_1 x_2$:

> s3:=product(x,x=T);

$$s3 := -\sqrt{3} + 1$$

+ Từ các kết quả trên, suy ra giá trị của Q:

> Q:=1/s1-s2/(3*s3);

$$Q := \frac{\sqrt{3}}{6} - \frac{\sqrt{2}}{3(-\sqrt{3} + 1)}$$

> Q:=simplify(Q,radical);

$$Q := \frac{3 - \sqrt{3} + 2\sqrt{2}}{6(\sqrt{3} - 1)}$$

> Q:=expand(Q,radical);

$$Q := \frac{\sqrt{3}}{6} + \frac{\sqrt{2}}{6} + \frac{\sqrt{2}\sqrt{3}}{6}$$

9. Hàm eval, evalf, value, allvalues. (các hàm tính giá trị)

a) Hàm value được sử dụng khi muốn tìm kết quả của các biểu thức như: Int (nguyên hàm, tích phân), Sum (tổng), Limit (giới hạn), Diff (đạo hàm).

Ví dụ 1:

Tính giá trị của biểu thức $F(x) = \int \left(x^2 + \frac{1}{x^2} \right) dx$.

+ Nhập biểu thức tích phân vào Maple:

> F(x) := Int(x^2+1/x^2, x);

$$F(x) := \int x^2 + \frac{1}{x^2} dx$$

+ Tìm một kết quả của phép toán trên (dùng lệnh 'value'):

> F(x):=value(%);

$$F(x) := \frac{x^3}{3} - \frac{1}{x}$$

Ví dụ 2:

Tìm giới hạn sau $G = \lim_{x \rightarrow 0} \frac{\sin 2x}{x}$.

+ Nhập biểu thức giới hạn vào Maple:

> G := Limit(sin(2*x)/x, x=0);

$$G := \lim_{x \rightarrow 0} \frac{\sin(2x)}{x}$$

+ Tính giá trị của giới hạn trên:

> **G:=value(G);**

$$G := 2$$

Ví dụ 3:

Tìm kết quả của tổng $S = 1 + 2 + \dots + n$.

+ Nhập biểu thức S:

> **S:=Sum(i,i=1..n);**

$$S := \sum_{i=1}^n i$$

+ Tìm kết quả của tổng trên:

> **S:=value(S);**

$$S := \frac{(n+1)^2}{2} - \frac{n}{2} - \frac{1}{2}$$

Có thể dùng gộp hai lệnh trên như sau:

> **S:=value(Sum(i,i=1..n));**

$$S := \frac{(n+1)^2}{2} - \frac{n}{2} - \frac{1}{2}$$

2) Hàm eval dùng để tính giá trị của một biểu thức với điều kiện cho trước.

Ví dụ 1:

Tính giá trị của biểu thức $P = 3x^2 - x - \sqrt{3}$ khi $x = \sqrt{3} + 1$; $x = -\frac{5}{2} + \sqrt{6}$.

+ Nhập biểu thức vào Maple:

> **f:=3*x^2-x-sqrt(3);**

$$f := 3x^2 - x - \sqrt{3}$$

+ Tính giá trị của f khi $x = \sqrt{3} + 1$:

> **eval(f,x=sqrt(3)+1);**

$$3(1 + \sqrt{3})^2 - 1 - 2\sqrt{3}$$

Làm gọn kết quả bằng lệnh 'radnormal':

> **radnormal(%);**

$$11 + 4\sqrt{3}$$

+ Tính giá trị của f khi $x = -\frac{5}{2} + \sqrt{6}$:

> **eval(f,x=-5/2+sqrt(6));**

$$3\left(-\frac{5}{2} + \sqrt{6}\right)^2 + \frac{5}{2} - \sqrt{6} - \sqrt{3}$$

> **radnormal**(%);

$$\frac{157}{4} - 16\sqrt{2}\sqrt{3} - \sqrt{3}$$

Ví dụ 2:

Tìm giá trị của biểu thức $f = x^2 + x - 1$ biết x là một nghiệm (thứ nhất) của phương trình $x^2 - x - 1 = 0$.

+ Xác định nghiệm thứ nhất của phương trình $x^2 - x - 1 = 0$ bằng lệnh:

> **n1:=RootOf(x^2-x-1,index=1);**
 $n1 := \text{RootOf}(_Z^2 - _Z - 1, \text{index} = 1)$

+ Tính giá trị của f khi $x = n1$:

> **eval(x^2+x-1,x=n1);**
 $\text{RootOf}(_Z^2 - _Z - 1, \text{index} = 1)^2 + \text{RootOf}(_Z^2 - _Z - 1, \text{index} = 1) - 1$

+ Hiện thị giá trị cụ thể của kết quả trên bằng lệnh ‘allvalues’:

> **allvalues**(%);

$$\left(\frac{\sqrt{5}}{2} + \frac{1}{2}\right)^2 + \frac{\sqrt{5}}{2} - \frac{1}{2}$$

Làm gọn kết quả trên:

> **radnormal**(%);

$$\sqrt{5} + 1$$

3. Hàm **evalf** dùng để tính giá trị gần đúng của một biểu thức, một phép toán. Kết quả hiển thị dưới dạng số thập phân.

Cú pháp: > **evalf(a, n);**

hoặc: > **evalf[n](a);**

Để hiển thị kết quả của a với n chữ số (bao gồm cả phần thập phân).

Ví dụ 1:

Tính giá trị gần đúng của $\sqrt{3}$ đến 17 chữ số thập phân:

+ Ta dùng lệnh trong Maple như sau:

> **evalf[18](sqrt(3));**
 1.73205080756887729

+ hoặc

> **evalf(sqrt(3),18);**
 1.73205080756887729

Ví dụ 2:

Tính giá trị gần đúng của số π :

+ Tính gần đúng đến 46 chữ số thập phân:

> **evalf[47](Pi);**
 3.141592653589793238462643383279502884197169396

+ Tính gần đúng đến 246 chữ số thập phân:

```
> evalf[247](Pi);
3.14159265358979323846264338327950288419716939937510582097494459230781
062862089986280348253421170679821480865132823066470938446095505821
725359408128481117450284102701938521105559644622948954930381964428
0975665933446128475648233786783165271202
```

✎ Ứng dụng lệnh này để minh họa số Pi là số vô tỉ (số thập phân vô hạn không tuần hoàn) khi dạy bài số thực cho học sinh. Giáo viên có thể lấy một vài số hữu tỉ và số vô tỉ và tính gần đúng giá trị của các số đó đến hàng trăm chữ số để học sinh quan sát và rút ra kết luận cần đạt được. Rõ ràng, máy tính cầm tay không thực hiện được công việc này.

* Chú ý: Nếu không khai báo số n thì Maple lấy kết quả mặc định với 10 chữ số (bao gồm cả phần thập phân).

Chẳng hạn:

```
> evalf(sqrt(1111));
33.33166662
```

Ví dụ 3:

Tìm nghiệm gần đúng của phương trình $x^2 - \sqrt{3}x - 7 = 0$.

+ Giải phương trình :

```
> eq:=x^2-sqrt(3)*x-7=0;
eq := x^2 - sqrt(3) x - 7 = 0
```

```
> solve(eq, {x});
```

$$\left\{ x = \frac{\sqrt{3}}{2} + \frac{\sqrt{31}}{2} \right\}, \left\{ x = \frac{\sqrt{3}}{2} - \frac{\sqrt{31}}{2} \right\}$$

+ Tính gần đúng nghiệm đến 13 chữ số (kể cả phần thập phân):

```
> evalf[13](%);
{x = 3.649907585200}, {x = -1.917856777630}
```

4) Hàm 'allvalues' được sử dụng khi muốn hiển thị các kết quả của một phép tính dưới dạng cụ thể (căn thức, lũy thừa, ...) và trực quan.

Ví dụ 1:

Tìm tập nghiệm của phương trình $x^4 - x - 1 = 0$.

+ Giải phương trình :

```
> Tn:=solve(x^4-x-1=0,x);
Tn := RootOf(_Z^4 - _Z - 1, index = 1), RootOf(_Z^4 - _Z - 1, index = 2),
RootOf(_Z^4 - _Z - 1, index = 3), RootOf(_Z^4 - _Z - 1, index = 4)
```

+ Hiển thị kết quả dưới dạng căn thức:

```
> allvalues([Tn]);
```

$$\begin{aligned}
 & \left[\frac{\sqrt{6} \sqrt{\frac{(108 + 12\sqrt{849})^{(2/3)} - 48}{(108 + 12\sqrt{849})^{(1/3)}}}}{12} + \left(\left(\right. \right. \right. \\
 & -6 \sqrt{\frac{(108 + 12\sqrt{849})^{(2/3)} - 48}{(108 + 12\sqrt{849})^{(1/3)}}} (108 + 12\sqrt{849})^{(2/3)} \\
 & + 288 \sqrt{\frac{(108 + 12\sqrt{849})^{(2/3)} - 48}{(108 + 12\sqrt{849})^{(1/3)}}} + 72 \sqrt{6} (108 + 12\sqrt{849})^{(1/3)} \left. \right) / \left(\right. \\
 & (108 + 12\sqrt{849})^{(1/3)} \sqrt{\frac{(108 + 12\sqrt{849})^{(2/3)} - 48}{(108 + 12\sqrt{849})^{(1/3)}}} \left. \right)^{(1/2)} / 12, \\
 & - \frac{\sqrt{6} \sqrt{\frac{(108 + 12\sqrt{849})^{(2/3)} - 48}{(108 + 12\sqrt{849})^{(1/3)}}}}{12} + \frac{1}{12} I \left(\left(\right. \right. \\
 & 6 \sqrt{\frac{(108 + 12\sqrt{849})^{(2/3)} - 48}{(108 + 12\sqrt{849})^{(1/3)}}} (108 + 12\sqrt{849})^{(2/3)} \\
 & - 288 \sqrt{\frac{(108 + 12\sqrt{849})^{(2/3)} - 48}{(108 + 12\sqrt{849})^{(1/3)}}} + 72 \sqrt{6} (108 + 12\sqrt{849})^{(1/3)} \left. \right) / \left(\right. \\
 & (108 + 12\sqrt{849})^{(1/3)} \sqrt{\frac{(108 + 12\sqrt{849})^{(2/3)} - 48}{(108 + 12\sqrt{849})^{(1/3)}}} \left. \right)^{(1/2)}, \\
 & \frac{\sqrt{6} \sqrt{\frac{(108 + 12\sqrt{849})^{(2/3)} - 48}{(108 + 12\sqrt{849})^{(1/3)}}}}{12} - \left(\left(\right. \right. \\
 & -6 \sqrt{\frac{(108 + 12\sqrt{849})^{(2/3)} - 48}{(108 + 12\sqrt{849})^{(1/3)}}} (108 + 12\sqrt{849})^{(2/3)} \\
 & + 288 \sqrt{\frac{(108 + 12\sqrt{849})^{(2/3)} - 48}{(108 + 12\sqrt{849})^{(1/3)}}} + 72 \sqrt{6} (108 + 12\sqrt{849})^{(1/3)} \left. \right) / \left(\right. \\
 & (108 + 12\sqrt{849})^{(1/3)} \sqrt{\frac{(108 + 12\sqrt{849})^{(2/3)} - 48}{(108 + 12\sqrt{849})^{(1/3)}}} \left. \right)^{(1/2)} / 12,
 \end{aligned}$$

$$\begin{aligned}
 & -\frac{\sqrt{6} \sqrt{\frac{(108+12\sqrt{849})^{(2/3)}-48}{(108+12\sqrt{849})^{(1/3)}}}}{12} - \frac{1}{12} I \left(\left(\right. \right. \\
 & 6 \sqrt{\frac{(108+12\sqrt{849})^{(2/3)}-48}{(108+12\sqrt{849})^{(1/3)}}} (108+12\sqrt{849})^{(2/3)} \\
 & \left. - 288 \sqrt{\frac{(108+12\sqrt{849})^{(2/3)}-48}{(108+12\sqrt{849})^{(1/3)}}} + 72 \sqrt{6} (108+12\sqrt{849})^{(1/3)} \right) \left. \right) \left(\right. \\
 & \left. (108+12\sqrt{849})^{(1/3)} \sqrt{\frac{(108+12\sqrt{849})^{(2/3)}-48}{(108+12\sqrt{849})^{(1/3)}}} \right)^{(1/2)} \left. \right)
 \end{aligned}$$

Nhận xét: Nghiệm (gồm cả nghiệm thực và phức) hiển thị quá dài dòng.

Ta dùng lệnh 'evalf' để xem kết quả gần đúng:

> **evalf(%);**

[1.220744085 -0.2481260627+ 1.033982062I, -0.7244919598
-0.2481260627- 1.033982062I]

Qua kết quả này ta thấy, phương trình có 2 nghiệm thực và 2 nghiệm phức.

Chú ý: Hàm 'allvalues' thường kết hợp với hàm 'RootOf' để liệt kê và hiển thị các nghiệm của một đa thức, biểu thức dưới dạng căn thức, lũy thừa.

10. Một số hàm lượng giác.

• Đổi đơn vị:

* *Đổi độ sang radian.*

Đổi góc có số đo α^0 sang 'radian'.

Cú pháp: > **convert(α *degrees, radians);**

Ví dụ 1:

+ Đổi số đo của góc 123^0 sang radian:

> **convert(123*degrees, radians);**
 $\frac{41\pi}{60}$

+ Đổi số đo của góc $\left(\frac{12}{23}\right)^0$ sang radian:

> **convert((12/23)*degrees, radians);**
 $\frac{\pi}{345}$

+ Đổi số đo của góc $1,27^0$ sang radian:

```
> convert(1.27*degrees, radians);  
0.007055555556  $\pi$ 
```

* Đổi radian sang độ.

Đổi góc có số đo a (radian) sang ‘độ’.

Cú pháp: > **convert(a , degrees);**

Ví dụ 2:

+ Đổi số đo của góc $\frac{15\pi}{7}$ (rad) sang ‘độ’:

```
> convert(15*Pi/7, degrees);  
 $\frac{2700 \text{ degrees}}{7}$ 
```

+ Đổi số đo của góc 5,23 (rad) sang ‘độ’:

```
> convert(5.23, degrees);  
 $\frac{941.40 \text{ degrees}}{\pi}$ 
```

• Một số hàm lượng giác thường gặp (trong chương trình toán phổ thông).

Các hàm sin, cosin, tang của một cung (góc) a được khai báo trong Maple với từ khóa tương ứng là :

```
> sin(a); > cos(a); > tan(a);
```

Trong đó cung (góc) a luôn được đo bởi đơn vị ‘radian’.

Ví dụ 3:

+ Để tính $\sin\left(\frac{2\pi}{3}\right)$, ta nhập vào Maple như sau:

```
> sin(2*Pi/3);  
 $\frac{\sqrt{3}}{2}$ 
```

Số ‘ π ’ trong Maple được khai báo bởi từ khóa ‘Pi’ hoặc ‘pi’. Nếu dùng ‘Pi’ thì Maple sẽ tính luôn giá trị của cung. Còn nếu dùng ‘pi’ thì sẽ thế nào, xin quý độc giả xem câu lệnh sau:

```
> sin(2*pi/3);  
 $\sin\left(\frac{2 \pi}{3}\right)$ 
```

Kết quả chỉ hiện thị dưới dạng biểu thức (kí hiệu). Do vậy quý độc giả cần chú ý khi sử dụng từ khóa của số π .

+ Tính $\tan\left(\frac{\pi}{12}\right)$:

```
> bt:=tan(Pi/12);  
 $bt := \tan\left(\frac{\pi}{12}\right)$ 
```

Kết quả hiển thị dưới dạng biểu thức. Để biểu diễn kết quả ở dạng cụ thể ta có thể dùng các hàm như sau:

- Biểu diễn kết quả dưới dạng thập phân dùng hàm 'evalf':

> **evalf(bt);**

$$0.2679491924$$

- Biểu diễn kết quả dưới dạng căn thức dùng hàm 'convert(bt, 'radical')':

> **convert(bt, 'radical');**

$$2 - \sqrt{3}$$

+ Tính $\cos(18^\circ)$:

Đầu tiên ta đổi đơn vị sang radian:

> **a:=convert(18*degrees, radians);**

$$a := \frac{\pi}{10}$$

Tiếp theo ta tính $\cos(a)$:

> **t:=cos(a);**

$$t := \cos\left(\frac{\pi}{10}\right)$$

Biểu diễn kết quả dưới dạng căn thức:

> **convert(t, 'radical');**

$$\frac{\sqrt{2} \sqrt{5 + \sqrt{5}}}{4}$$

11. Biến đổi các biểu thức lượng giác.

a) Dùng hàm 'simplify(expr, trig)' để đơn giản một biểu thức 'expr' dạng lượng giác. Nếu biểu thức 'expr' là một đa thức theo $\sin(x)$ và $\cos(x)$ thì kết quả của lệnh trên là một biểu thức dạng chứa các lũy thừa của $\sin(x)$ và $\cos(x)$ đã được rút gọn sử dụng kết quả $\sin^2 x + \cos^2 x = 1$.

Kết quả thường có dạng: $\sin^n x \cdot \cos^m x \cdot (A \cdot \sin x + B)$, trong đó A, B là các đa thức theo $\cos(x)$.

Ví dụ 1:

Biểu diễn biểu thức $A = \cos 2x - \sin^2 x$ theo $\sin(x)$ và $\cos(x)$:

+ Nhập biểu thức vào Maple:

> **A:=cos(2*x)-sin(x)^2;**

$$A := \cos(2x) - \sin(x)^2$$

+ Dùng lệnh trên ta được kết quả:

> **A:=simplify(A, trig);**

$$A := 3 \cos(x)^2 - 2$$

+ Nếu muốn phân tích $\cos 2x$ (dùng công thức nhân đôi) ta dùng lệnh:

```
> A:=simplify(A,[cos(2*x)]);
```

$$A := 2 \cos(x)^2 - 1 - \sin(x)^2$$

Ví dụ 2:

Biến đổi biểu thức $g = \sin x \cdot \cos^2 x - \cos^3 x + \cos x$ thành tích.

+ Nhập biểu thức vào Maple:

```
> g := sin(x)*cos(x)^2-cos(x)^3+cos(x);
```

$$g := \sin(x) \cos(x)^2 - \cos(x)^3 + \cos(x)$$

+ Dùng lệnh ‘simplify_trig’ để biến đổi:

```
> g:=simplify(g,trig);
```

$$g := \sin(x) \cos(x) (\cos(x) + \sin(x))$$

Ví dụ 3:

Biến đổi biểu thức sau về dạng tích: $f = \sin x + \sin 2x + \sin 3x$.

+ Đối với các phép biến đổi lượng giác chúng ta cần khéo léo sử dụng các hàm một cách phù hợp để được kết quả như ý muốn. Trong ví dụ này tôi muốn phân tích để quý độc giả thấy được hiệu lực của câu lệnh có tác dụng như thế nào !

+ Nhập biểu thức vào Maple:

```
> f:=sin(x)+sin(2*x)+sin(3*x);
```

$$f := \sin(x) + \sin(2x) + \sin(3x)$$

+ Nếu quý độc giả dùng hàm ‘factor’ để biến đổi biểu thức trên thì liệu kết quả thu được sẽ như thế nào. Hãy xem lệnh sau:

```
> f:=factor(f);
```

$$f := \sin(x) + \sin(2x) + \sin(3x)$$

Chúng ta nhận thấy, kết quả vẫn không có gì thay đổi.

+ Nếu quý độc giả dùng lệnh ‘simplify_trig’ thì liệu kết quả sẽ thế nào ?

```
> f:=simplify(f,trig);
```

$$f := \sin(x) + \sin(2x) + \sin(3x)$$

Ồ! Kết quả vẫn cứ trơ như vậy.

+ Theo ý tôi, đầu tiên tôi phân tích biểu thức theo $\sin(2x)$:

```
> f:=simplify(f,[sin(2*x)]);
```

$$f := 2 \sin(x) \cos(x) + 4 \sin(x) \cos(x)^2$$

Kết quả thu được đã có sự thay đổi. Bây giờ có thể dùng lệnh ‘factor’ hoặc ‘simplify_trig’ để biến đổi thành tích:

```
> f:=simplify(f,trig);
```

$$f := 2 \sin(x) \cos(x) (1 + 2 \cos(x))$$

Hoặc:

```
> f:=factor(f);
```

$$f := 2 \sin(x) \cos(x) (1 + 2 \cos(x))$$

Vậy kết quả đã có được như ý muốn !

b) Dùng hàm 'combine(expr, trig)' để hạ bậc một biểu thức lượng giác 'expr'.

Hàm này sử dụng các công thức biến đổi tích thành tổng để hạ bậc tối đa cho một biểu thức lượng giác chứa lũy thừa.

Ví dụ 4:

Hạ bậc biểu thức sau: $Q = \sin^4 x + \cos^4 x$.

+ Nhập biểu thức vào Maple:

> **Q:=sin(x)^4+cos(x)^4;**

$$Q := \sin(x)^4 + \cos(x)^4$$

+ Hạ bậc Q dùng lệnh 'combine_trig':

> **Q:=combine(Q, trig);**

$$Q := \frac{3}{4} + \frac{1}{4} \cos(4x)$$

Ví dụ 5: (Bài tập 46b/tr 215 SGK ĐS10 nâng cao)

Chứng minh rằng: $\sin x \sin\left(\frac{\pi}{3} - x\right) \sin\left(\frac{\pi}{3} + x\right) = \frac{1}{4} \sin 3x$.

+ Nhập vế trái vào Maple:

> **Vt:=sin(x)*sin(Pi/3-x)*sin(Pi/3+x);**

$$Vt := \sin(x) \cos\left(\frac{\pi}{6} + x\right) \sin\left(\frac{\pi}{3} + x\right)$$

+ Rút gọn vế trái dùng lệnh 'combine_trig':

> **Vt:=combine(b, trig);**

$$Vt := \frac{1}{4} \sin(3x)$$

Kết quả là VT = VP (đpcm).

Ví dụ 6: (Bài tập 50/tr215 SGK ĐS10 nâng cao)

Chứng minh biểu thức sau không phụ thuộc vào x:

$$\cos^2(a+x) + \cos^2 x - 2 \cos a \cdot \cos x \cdot \cos(a+x)$$

+ Nhập biểu thức sau vào Maple:

> **restart;**

Q:=cos(a+x)^2+cos(x)^2-2*cos(a)*cos(x)*cos(a+x);

$$Q := \cos(a+x)^2 + \cos(x)^2 - 2 \cos(a) \cos(x) \cos(a+x)$$

+ Dùng hàm 'combine_trig' để rút gọn:

> **Q:=combine(Q, trig);**

$$Q := \frac{1}{2} - \frac{1}{2} \cos(2a)$$

Kết quả của biểu thức (sau khi rút gọn) không chứa x. Đó là điều cần chứng minh.